

Paper 389-2012

Make an appropriate page break of PDF when using PROC REPORT

Xia Ke Shan , Chinese Financial Electrical Company, Beijing, China

ABSTRACT

When using proc report to generate a PDF file, SAS® will not split two group values if the current page can't hold the next group value any more. What we need is that when the current page can't hold the next group value we want to push the next group value into the next page (i.e. make a page break at appropriate position). This paper is to try to solve this problem. SAS® introduced the SPANROWS option in SAS® 9.2 for PROC REPORT which does address this issue, but what if you don't have SAS® 9.2 or you don't like the way the spanned rows looks on your report. Then you still might benefit from learning a technique to take control of page breaking.

INTRODUCTION

When using proc report to generate a PDF file, SAS® can't make a page break at appropriate position in a PDF file. Let me explain this question more detail.

Assuming after running the following code

```
ods listing close;
    options papersize=A4 nodate;
    options orientation=landscape leftmargin="0.2cm" rightmargin="0.2cm"
topmargin="0.5cm" bottommargin="0.5cm";
    goptions ftext = "helvetica/bold" noborder device=CGMOF97P NOGRAPHRC;
    ods pdf style=statdoc file='c:\samplereport.pdf' notoc style=statdoc;
    title 'Sample Report';
proc report data=test.test nowd headline ;
define manager / order order=formatted;
run;
    ods pdf close;
    ods listing;
```

The PDF file looks like:

Make an appropriate page break of PDF when using PROC REPORT, Continued

Sample Report

Manager	Department	Sales
Adams	Canned	225
	Meat/Dairy	350
	Paper	40
	Produce	80
Alomar	Canned	420
	Meat/Dairy	190
	Paper	90
	Produce	88
Andrews	Canned	420
	Meat/Dairy	300
	Paper	200
	Produce	125
Brown	Canned	230
	Meat/Dairy	250
	Paper	45
	Produce	73
Jones	Canned	220
	Meat/Dairy	300
	Paper	40
	Produce	70
Pelfrey	Canned	420
	Meat/Dairy	205
	Paper	45
	Produce	78
Reveiz	Canned	420
	Meat/Dairy	600
	Paper	60
	Produce	30
Smith	Canned	120

Sample Report

	Meat/Dairy	100
	Paper	50
	Produce	80
Taylor	Canned	120
	Meat/Dairy	130
	Paper	53
	Produce	50

We will see there is a Smith at the bottom of the first page, but we want all of Smith to move to the next page, if the current page can not hold all the Smith. That is to say PDF should look like:

Make an appropriate page break of PDF when using PROC REPORT, Continued

Desired Report

Manager	Department	Sales
Adams	Canned	225
	Meat/Dairy	350
	Paper	40
	Produce	80
Alomar	Canned	420
	Meat/Dairy	190
	Paper	90
	Produce	88
Andrews	Canned	420
	Meat/Dairy	300
	Paper	200
	Produce	125
Brown	Canned	230
	Meat/Dairy	250
	Paper	45
	Produce	73
Jones	Canned	220
	Meat/Dairy	300
	Paper	40
	Produce	70
Pelfrey	Canned	420
	Meat/Dairy	205
	Paper	45
	Produce	78
Reveiz	Canned	420
	Meat/Dairy	600
	Paper	60
	Produce	30

Desired Report

Manager	Department	Sales
Smith	Canned	120
	Meat/Dairy	100
	Paper	50
	Produce	80
Taylor	Canned	120
	Meat/Dairy	130
	Paper	53
	Produce	50

How can we get? First of all ,we need to count the number of lines in a page, and then make a flag variable to identify when to break at one page. If the number of a group value were greater than the number of lines in a page, SAS® will automatically break it at appropriate position .we only need to consider the mod of the number of a group value, if the sum of these mod at the first observation of a group then we need to break (i.e. the flag variable should add one by itself).

The following code is worked under SAS® 9.2

Make an appropriate page break of PDF when using PROC REPORT, Continued

```
data test;
infile datalines expandtabs;
input Manager : $20. Department : $20. Sales ;
datalines;
Adams Canned 225
Adams Meat/Dairy 350
Adams Paper 40
Adams Produce 80
Alomar Canned 420
Alomar Meat/Dairy 190
Alomar Paper 90
Alomar Produce 86
Andrews Canned 420
Andrews Meat/Dairy 300
Andrews Paper 200
Andrews Produce 125
Brown Canned 230
Brown Meat/Dairy 250
Brown Paper 45
Brown Produce 73
Jones Canned 220
Jones Meat/Dairy 300
Jones Paper 40
Jones Produce 70
Pelfrey Canned 420
Pelfrey Meat/Dairy 205
Pelfrey Paper 45
Pelfrey Produce 76
Reveiz Canned 420
Reveiz Meat/Dairy 600
Reveiz Paper 60
Reveiz Produce 30
Smith Canned 120
Smith Meat/Dairy 100
Smith Paper 50
Smith Produce 80
Taylor Canned 120
Taylor Meat/Dairy 130
Taylor Paper 53
Taylor Produce 50
;
run;
/*to test the page size of pdf*/
ods listing close;
```

Make an appropriate page break of PDF when using PROC REPORT, Continued

```

options papersize=A4 nodate;
options orientation=landscape leftmargin="0.2cm" rightmargin="0.2cm"
topmargin="0.5cm" bottommargin="0.5cm";
options ftext = "helvetica/bold" noborder device=CGMOF97P NOGRAPHRC;
ods pdf style=statdoc file='c:\samplereport.pdf' notoc style=statdoc;
title 'Sample Report';
proc report data=test nowd headline ;
define manager / order order=formatted;
run;

ods pdf close;
ods listing;

/*After testing it, found a page contains 29 lines*/

proc means data=test nway noprint;
class manager;
output out=count n=count;
run;
data result;
merge test count(keep=manager count);
by manager;
run;
data result;
set result;
mod=mod(count,29);
run;
data result; * To decide Page break point;
set result;
by manager;
retain break 1;
if last.manager then sum_mod+mod;
if count +sum_mod ge 29 and manager ne lag(manager) then do;
                    break+1;
                    sum_mod=0;
                    end;
run;

/*get the desired pdf */
ods listing close;
options papersize=A4 nodate;
options orientation=landscape leftmargin="0.2cm" rightmargin="0.2cm"
topmargin="0.5cm" bottommargin="0.5cm";

```

Make an appropriate page break of PDF when using PROC REPORT, Continued

```
options ftext = "helvetica/bold" noborder device=CGMOF97P NOGRAPHRC;
ods pdf style=statdoc file='c:\desired_report.pdf' notoc style=statdoc;
title 'Desired Report';
proc report data=result nowd headline ;
column break manager department sales ;
define break /group noprint;
define manager / order order=formatted;
break after break /page;
run;

ods pdf close;
ods listing;
```

CONCLUSION

Anyway, if folks don't want to use SPANROWS because they don't like the way it looks or they don't have SAS® 9.2, then this technique is benefited.

.

AUTHOR CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Xia Ke Shan

JINRONG STREET, XICHENG DISTRICT BEIJING

Email: xiakeshan@yahoo.com.cn

SAS® and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.