

## Paper 263-2012

**Faking Esri ArcGIS maps in SAS®**

Anastasiya Osborne, Farm Service Agency (USDA), Washington, DC

**ABSTRACT**

This paper describes another successful project of automating manual tasks at the Farm Service Agency, USDA. After an urgent request, new SAS® code with PROC GPROJECT, PROC GMAP, and heavy use of annotation was developed to map widely used state revenue guarantees and benchmark yields for 22 crops under the Congressionally-mandated Average Crop Revenue Election (ACRE) program. A program is administered by the Farm Service Agency, USDA. Previously, such maps were created in ArcGIS. The SAS solution enhanced the quality of the map: it showed state revenues and yields in a compact but readable way, drastically improved presentation of data in Northeast states, and eliminated repetitive task of making a PDF file out of 22 separate maps.

**INTRODUCTION**

As mentioned in an earlier paper (4), being a member of the Economic and Policy Analysis Staff at the Farm Service Agency (FSA), USDA requires stamina and creativity. The office is responsible for creating analytic reports for inside and outside audiences (senior management and policy makers). Producing those reports can be time-consuming and inefficient. This is when the ability to automate tasks in SAS becomes invaluable. Such automation takes some time investment in the beginning of a project, and could look like a waste of time to some, but an automated process ultimately reduces turn-around time of regular requests and fool-proofs the majority of the steps.

After the sudden departure of a colleague in March 2011, the author of the paper volunteered to pick up some of his previous tasks. One of them was the bi-monthly creation of a PDF file with 22 maps of regularly updated administrative data for major field crops, to make the outcomes of ACRE, a major Congressionally-mandated program, accessible on the FSA website. This task used to be accomplished in ArcGIS. It required laborious manual work in Excel to redo links to modified data sources twice a month, daunting manual data manipulations with shapefiles in ArcGIS, and subsequent tedious task of combining maps in one PDF file in a certain order. The author was unfamiliar with ArcGIS at the time and managed to have it installed only after the new SAS maps were already made and put on the USDA website. Fortunately, she knew how to map using SAS – and took upon herself to simulate maps in SAS so that no one would know how they were created. The goal of the project was not necessarily to make the most appealing user-friendly maps to display the relevant data, but to retain the familiar appearance of the maps that previously had been generated with ArcGIS, in case FSA website users had grown accustomed to them over recent years. This suggested the informal title of the project - the author was “faking the maps.”

The resulting SAS program:

- automatically reads Excel files from a shared drive, using options LIBNAME xlsdta EXCEL, HEADER=YES, MIXED=YES;
- converts maps.uscenter data from longitude and latitude of the center of each state to X and Y;
- creates several annotation datasets, to make
  1. a fake Alaska textbox,
  2. a fake legend on the bottom, and
  3. labels with different font colors to display either two blue values for all revenue and all yield, or four red and black values for irrigated and non-irrigated revenue and yield;
- uses %annomac;
- calls a module that creates custom-made labels for data-heavy or small Northeast states without overcrowding of the map area;
- uses GOTO command to circumvent the code flow for the most complicated labels in Delaware;
- significantly saves project time and improves data presentation.

## Faking Esri ArcGIS maps in SAS®, continued

### THE MAPS IN ARCGIS AND SAS


The PDF file with 22 maps is located on the Farm Service Agency website, <http://www.fsa.usda.gov/FSA/webapp?area=home&subject=dccp&topic=landing>. It is called "2011 State Guarantee and Benchmark Yield Maps," and the most recent as of this writing was updated on June 10, 2011 (see Fig. 2.)

There are 22 crops that are included in ACRE: <sup>1</sup> Major commodities, such as corn, soybeans, and wheat, are more widely grown, and thus have a substantial impact on agriculture. Generating maps for those crops is more challenging, because of the larger numbers (more digits, separated by commas) for production and yield, requiring careful labeling to ensure readability. Moreover, the difficulty is further compounded by the fact that different crops use different units of measure. That is, yields are expressed in **bushels** for barley, corn, flaxseed, grain sorghum, oats, soybeans, and wheat (major crops); and **pounds** for the other crops.

Below (Fig. 1.) is a print-screen of the Farm Service Agency page about ACRE program. The link to the Direct and Counter-Cyclical Program/ACRE information is on

<http://www.fsa.usda.gov/FSA/webapp?area=home&subject=dccp&topic=landing>.

To learn more about the specifics of the program, go to [http://www.fsa.usda.gov/Internet/FSA\\_File/acrebkgrd.pdf](http://www.fsa.usda.gov/Internet/FSA_File/acrebkgrd.pdf) and <http://www.ers.usda.gov/Briefing/farmpolicy/acre.htm>.



<sup>1</sup> The 22 crops included in the Average Crop Revenue Election program are: barley, canola, corn, crambe, dry peas, flaxseed, grain sorghum, large chickpeas, lentils, large rice, medium rice, mustard, oats, peanuts, rapeseed, safflower, sesame, small chickpeas, soybeans, sunflowers, upland, cotton, and wheat.

## Faking Esri ArcGIS maps in SAS®, continued

Figure 1. FSA page about ACRE.

The screenshot shows the FSA ACRE website. On the left is a navigation menu with links: Laws and Regulations, Outreach and Education, Payment Eligibility, Price Support, and Tobacco. The main content area is titled 'The following files provide ACRE National, State and County data'. It lists data for 'Program Year 2011' and 'Program Year 2010'. A red box labeled 'Latest data' points to the '2011 State Guarantee and Benchmark Yield Maps (PDF; 1.6 MB; June 10, 2011)' link. To the right of the main content are sections for 'DCP Temporary Directives', 'Email Updates', 'RSS Feeds', 'FSA Widgets', 'Ask FSA', and 'Media Help'.

**Navigation Menu:**

- ▶ Laws and Regulations
- ▶ Outreach and Education
- ▶ Payment Eligibility
- ▶ Price Support
- ▶ Tobacco

**The following files provide ACRE National, State and County data**

**Program Year 2011**

- 2011 ACRE Prices Values (XLS; 26 KB; June 10, 2011)
- Preliminary 2011 ACRE Benchmark Parameters (XLS; 133 KB; June 10, 2011)
- 2011 Program Year County Yields (XLS; 2.86 MB; May 20, 2011)

(Note that the file contains "All" Yields for barley, corn, grain sorghum, oats, soybeans, and wheat. Check back to the website for file updates.)

**Program Year 2010**

- 2010 ACRE Prices Values (XLS; 27 KB; June 10, 2011)
- Preliminary 2010 Acre Benchmark Parameter Values (XLS; 134 KB; April 15, 2011)
- 2010 State Guarantees and Benchmark Yields (PDF 1.5 MB; April 15, 2011)
- 2010 Program Year County Yields (XLS; 4.06 MB; April 15, 2011)
- 2010 Scheduled Releases ACRE Data (XLS; 26 KB; April 9, 2010)

**DCP Temporary Directives**

- Farm Security and Rural Investment Act of 2002 (2002 Farm Bill) (Pub. L. 107-171, May 12, 2002)
- 7 CFR 1412 Direct and Counter-Cyclical Program and Peanut Quota Buyout Program

**Email Updates**

**RSS Feeds**

**FSA Widgets**

**Ask FSA**

**SHARE**

**Media Help**

To view PDF files you must have Adobe Acrobat Reader installed on your computer.

To view Flash files you must have Macromedia Flash Player installed on your computer.

Figure 2. Continuation of the FSA page about ACRE – where the downloadable maps files are located.


The maps on FSA website and the source data for them, ACRE Price Values and ACRE Benchmark Parameters, are updated once or twice a month.

Fig. 3 (on p.4) presents a corn map made in ArcGIS in October 2010. Corn is the most difficult crop to map, because of the sheer production size – every single state in the United States except Alaska grows it. There are 10 states: six grouped together (NE, CO, KS, OK, TX, and LA) in the middle of the map, and three bordering the Atlantic ocean (DE, GA, and FL), that report not only all revenue and all yield (in blue), but also more detailed data, irrigated revenue and irrigated yield (in red), plus non-irrigated revenue and non-irrigated yield (in black.) Red-black combination happens when at least  $\frac{1}{4}$  of planted acres is irrigated and  $\frac{1}{4}$  is non-irrigated. Irrigation makes it possible to produce larger yields of better quality. No wonder revenues for irrigated crops can be up to 4 times higher than for non-irrigated crops. That makes it harder for a programmer to design SAS code that would allow labels to vary in color and length, but still look good on a map for any crop, any combination of values, and in any state.

A closer look at Fig. 3 tells us that state abbreviations and values for many Northeastern and Mid-Atlantic states are moved off the U.S. map and into the ocean. The previous analyst who mapped in ArcGIS solved the problem of putting busy labels with four numbers for small states by placing textboxes in the ocean. This is the case for RI, VT, CT, NJ, MD, DE, and even FL. It is likely that such placement was the most convenient solution in ArcGIS because the textboxes would be too large to position within map boundaries for those states. However, even if we accept such ocean placement, another problem is that textboxes are not located directly across the represented states, making it difficult to read the map. DE extends to NC. VT, NH, and RI are not where they need to be.

Fig. 4 (on p.4 as well, for better comparison) shows the SAS solution, with 2011 data. It is easy to see that there is a way to present data in Northeastern and Mid-Atlantic states without using too much of the ocean area. Blue, red, and black labels with the state guarantees and yields are placed directly on each state area in a compact but readable way. Instead of four lines for irrigated/non-irrigated values, only two are needed: one line for red, one for black. This is accomplished by some tweaking - some labels are reduced in size for smaller states, and moved to the right or left from the standard position as needed.


# **Corn Final ACRE Program Guarantees in (\$/Acre) and Benchmark State Yields in (Bu./Acre) for 2010 Crop Year** **\$3.81 Guarantee Price in (\$/Bu.)**





# Corn Preliminary ACRE Program Guarantees in (\$/Acre) and Benchmark State Yields in (Bu./Acre) for 2011 Crop Year

\$4.45 Guarantee Price in (\$/Bu.)



Latest Update: June 15, 2011

## Faking Esri ArcGIS maps in SAS®, continued

### NORTHEASTERN AND MID-ATLANTIC STATES

The Northeastern States include the New England states of Maine, New Hampshire, Vermont, Massachusetts, Connecticut, and Rhode Island, that are more of interest to us because they are small, making it hard to put textboxes containing anything else than state abbreviations. They also include New York, northern New Jersey, and northern Pennsylvania. Mid-Atlantic States include New Jersey, Maryland, and Delaware. Delaware, "The First State," gets the prize for having the most difficult textbox to work with, especially in case of corn - because the state area is just too small for four values of irrigated and non-irrigated revenue and yield, and also because labels for New Jersey and Maryland tend to crowd the area in maps for crops with large production (See Fig. 5.)

All small ocean states have agricultural production participating in ACRE. The SAS program "AO Supermodule general - revenue and yield.sas" deals with crowded states by moving the position of the labels.




Figure 5. Northeastern and Mid-Atlantic states with small states, difficult to map. (1)

The SAS Technical note in SAS/GRAPH(R) 9.2: Reference, Second Edition (5), about placement and alignment of a text string specified by the LABEL function, was very helpful in figuring out the ways to place labels directly on the map without moving them to the ocean.




Figure 6. Different text positions in the labels to ensure readability of yield and revenue for each crop.

## Faking Esri ArcGIS maps in SAS®, continued

Ideally, to be most pleasing to the eye, text strings in a label for each state should be centered, as in (Fig.6. a). However, this general idea does not work well with the U.S. map, due to the differences in the state areas, irrigation practices, and resulting yield and revenues. Originally, sorghum, soybean, and sunflower maps had problems with DE and MA. Corn, wheat, and oat maps had problems with NH, MA, RI, CT, MD, DE, and DC. Those states needed to be labeled differently. Just how much differently - the author had to find out by trial and error, until text strings were possible to read for any crop in any state.

In the end, labels for NH, RI, and DE were moved to the right (Fig.6. b), labels for CT and MA were moved to the left (Fig.6. c), to make the labels for other states readable.

### FAKE ALASKA TEXTBOX WITH VALUES AS MACRO VARIABLES

Only three ACRE crops out of 22 have values for Alaska: barley, oat, and wheat. However, the values had to be shown, and a special textbox accomplished the task. The textbox for Alaska is fake and does not resemble in any way the geographically challenging outline map for the "Land of the Midnight Sun," as in Fig.7. It took a series of tweaking to correctly position the textbox on the U.S. map and put labels with the state abbreviation, all revenue, and all yield in that textbox.



Figure 7. Outline map of Alaska.




Figure 8. Fake Alaska on the top left corner of the U.S. map, from the wheat map as of June 15, 2011.

The code for making this textbox is below. It creates global macro variables for Alaska's All Yield and All Revenue from the annotated dataset anno8, using CALL SYMPUT. Alaska is not big on crop production, and having any crop value is surprising. Only two blue values show up for only three crops, and there is no need to worry about more complicated SAS code that would create red/black labels for showing theoretically possible but never reported irrigated/non-irrigated values – at least for now.

After a data step creating macro variables, the SAS Tech Support –created program %annomac is used, to make life of SAS/GRAPH users easier. %RECT(-3,85,3,92,black,1,2) draws the textbox outline, and macro variables serve as labels for All Revenue at y = 89, and All Yield at y = 87.

## Faking Esri ArcGIS maps in SAS®, continued

```

data forAK ;
  set anno8 ;
  if state = 2 ;
  AK_&CROP.A = cat(&CROP.A); /* A - all revenue */
  AK_&CROP.AY = cat(&CROP.AY); /* AY - all yield */

  call symput("AK_Allrev", AK_&CROP.A);
  call symput("AK_Allyld", AK_&CROP.AY);
run ;
/* Make an annotate data set to add Alaska box in the upper left. First, use %ANNOMAC macro call, to
make annotate macro available. */

%annomac;

/* Now - use those macro variables for Alaska. */
data alaska;
  length text $20;
  retain xsys ysys '5' hsys '3' ;
  %rect(-3,85,3,92,black,1,2); /* to create a textbox */
  function = 'label';
  size = 1.25;
  x = 0;
  y = 91;
  text = 'AK';
  output;

  color = 'blue';
  x = 0;
  y = 89;
  text = "&AK_Allrev";
  output ;

  x = 0;
  y = 87;
  text = "&AK_Allyld";
  output;
run;

data anno ;
  set anno8 (where = (state ^= 2)) ;
run ;

```

### READING EXCEL FILE INTO SAS:

The online proceedings from the SAS Global Forum 2011 in Las Vegas became available even before the conference itself took place in April 2011, just in time for the author to take a look at Benjamin Jr's paper (2). The paper helped to come up with a way to read pesky Excel files, using options LIBNAME xlsx EXCEL, HEADER=YES, MIXED=YES. The problem with the input data was that most of the variables were numeric in Excel, but corn and wheat values were character. Also, Economic and Policy Analysis management desired to see formatted values on the maps, with revenue reported as X,XXX.DD and yields as XXX.D. Comma signs made it harder to process data with more elegant arrays in one program. In the beginning of the project, there were three programs, one elegant and two bland down-to-business ones for special cases of corn and wheat. At the end, one longer general SAS program served the purpose. The code was less elegant but worked just fine, converting every variable of interest into a character one and dealing with numerous special cases for crop-state combinations. The MIXED option is needed when the Excel columns have both character and numeric values. MIXED=YES reads all the



## Faking Esri ArcGIS maps in SAS®, continued

specified variables as character. However, even with that option the author had to input the variables as character, using COMPRESS(INPUT(&CROP.I, \$CHAR15.)).

```
LIBNAME xlsdta EXCEL "C:\My NESUG 2011 - faking\2011 Revenue as of June 10
2011.xls" HEADER=YES MIXED=YES;
```

```
Data test1 (rename = (new&CROP.A = &CROP.A new&CROP.AY = &CROP.AY
                      new&CROP.I = &CROP.I   new&CROP.IY = &CROP.IY
                      new&CROP.N = &CROP.N   new&CROP.NY = &CROP.NY)) ;
length state 5. abbr $2. ;
Set xlsdta."Sheet1$"n (drop = state);
```

```
state = input(code, 5.) ;
abbr = fipstate(state);
if state = . then delete ;
```

```
new&CROP.A = input(&CROP.A,$char15.);
old&CROP.A = &CROP.A ;
...
```

```
new&CROP.NY = input(&CROP.NY,$char15.);
old&CROP.NY = &CROP.NY ;
```

```
drop &CROP.A &CROP.AY &CROP.I &CROP.IY &CROP.N &CROP.NY ;
```

```
Run;
```

### THE PROGRAMS:

There are two programs. The short master program, "Crops all over.sas" runs the program "Supermodule general - revenue and yield.sas" for 22 crops. The parameters that need to be updated are the date of the run, the input file location, and **GUARPRICE** – the ACRE Price Guarantee (\$ per unit), that slightly changes at each update. There were plans to create macro variables for each ACRE Price Guarantee as parameters from a separate Excel file. However, experience showed that it is safer for now to keep this step manual, to make sure that the links in the input Excel files (given to the author) are set up correctly and the values indeed change from the previous run. See Appendix 2 to get acquainted with the look of the ACRE Price guarantees, updated in a master program "Crops All Over.sas."

The module:

- converts maps.uscenter data from longitude and latitude of the center of each state to X and Y;
- creates several annotation datasets, to make
  1. a fake Alaska textbox,
  2. a fake legend on the bottom, and
  3. labels with different font colors to display either two blue values for all revenue and all yield, or four red and black values for irrigated and non-irrigated revenue and yield;
- uses %**annomac**;
- uses GOTO command to circumvent the code flow for the most complicated labels in Delaware;
- significantly saves project time and improves data presentation.

See Appendix 1 for the abbreviated code of "Crops all over.sas" and Appendix 3 a full code of "Supermodule general - revenue and yield.sas." The programs are carefully commented, and only the size of them prompted to move them to Appendix section.

### CONCLUSION

This paper shows how to SAS-automate the creation of maps that were previously done in ArcGIS. The unexciting set of manual tasks from updating links in Excel file to working with shapefiles in ArcGIS to making a PDF file presents a challenge for an efficiency-minded SAS programmer. However, it can be automated in a short period of time. The end result is rewarding. The SAS solution shows the data in a compact but readable way, uses custom

## Faking Esri ArcGIS maps in SAS®, continued

labeling to present data even in tight spots like Northeast states, and automatically makes a PDF file with 22 maps. The quality of the map is improved, and both the process time and opportunity for mistakes are drastically reduced.

### REFERENCES

1. Source of the map in Fig. 5.: <http://www.americanbannorexchange.com/us-northeast.gif>
2. Benjamin Jr, William E. "The Little Engine That Could: Using LIBNAME Engine Options to Enhance Data Transfers Between SAS and Microsoft Excel Files." SAS Global Forum 2011. Available at: <http://support.sas.com/resources/papers/proceedings11/076-2011.pdf>
3. Huffer, Hillary. "Integrating SAS® and ArcMap: A Case Study of European Internet Use and Population." SAS Global Forum 2011. Available at: <http://support.sas.com/resources/papers/proceedings11/224-2011.pdf>
4. Osborne, Anastasiya. "Let Me Look At It! Graphic Presentation of Any Numeric Variable." SAS Global Forum 2009. Available at: <http://support.sas.com/resources/papers/proceedings09/072-2009.pdf>
5. The SAS Technical Note in SAS/GRAPH(R) 9.2: Reference, Second Edition.  
[http://support.sas.com/documentation/cdl/en/graphref/63022/HTML/default/viewer.htm#annotate\\_position.htm](http://support.sas.com/documentation/cdl/en/graphref/63022/HTML/default/viewer.htm#annotate_position.htm)
6. Zdeb, Mike. 2002. *Maps Made Easy Using SAS*. Cary, NC: SAS Institute Inc.  
SAS Institute Inc. 2004, *SAS/GRAPH® Software: Reference, Version 9*, Cary, NC: SAS Institute Inc.

### ACKNOWLEDGMENTS

My warmest thanks and gratitude extend to Mike Zdeb, SAS/GRAPH® Map Expert, whose help was invaluable.

Thanks to Joy Harwood, the Economic and Policy Analysis Staff director at the Farm Service Agency, USDA; and to Terry Hickenbotham, my supervisor at the Farm Loan Analysis Group, for allowing me to work on and present this paper at SAS meetings and conferences.

### CONTACT INFORMATION

Your comments and questions are valued and highly encouraged. Contact the author at:

Anastasiya Osborne

Farm Service Agency, USDA

1400 Independence Ave, S.W.

Washington, DC 20250-0506

Work Phone: 202-690-0446.

E-mail: [Anastasiya.Osborne@wdc.usda.gov](mailto:Anastasiya.Osborne@wdc.usda.gov)

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.

## Faking Esri ArcGIS maps in SAS®, continued

### APPENDIX 1. MASTER PROGRAM “AO CROPS ALL OVER.SAS”

```

/*****
/* Program: AO Crops all over.sas */
/* Input data: */
/* S:\EPAS\2008FBIMP\ACRE Revenue Guarantees\2011_mmdd\Revenue.xls */
*****/

Options symbolgen mlogic mprint mfile;

libname ps "C:\My NESUG 2011 - faking\";
%let lryear=2011;
%let mydate = June 17, 2011;

filename OUT "C:\My NESUG 2011 - faking\faking 2011 State Guarantees and Benchmark Yields as
of &mydate..pdf" ;

LIBNAME xlsdta EXCEL "H:\FSA SAS programs\Phil Sronce - March 2011\2011 Revenue as of June
10 2011.xls"

          HEADER=YES
          MIXED=YES;

/* 1. Barley */
%macro SUPERMODULE (CROP = Barley , CR = Barley, state_yield = Bu./Acre,
guarprice = $2.81, unit = $/Bu. ,
OUT = "C:\My NESUG 2011 - faking\faking 2011 State Guarantees and Benchmark Yields as of
&mydate..pdf" ) ;
%include "C:\My NESUG 2011 - faking\Supermodule general - revenue and yield.sas";
%mend SUPERMODULE ;
%SUPERMODULE (CROP = Barley , CR = Barley, state_yield = Bu./Acre, guarprice
= $2.81, unit = $/Bu. ,
OUT = "C:\My NESUG 2011 - faking\faking 2011 State Guarantees and Benchmark Yields as of
&mydate..pdf" );

...

/* 22. Wheat */
%SUPERMODULE (CROP = Wheat, CR = Wheat, state_yield = Bu./Acre, guarprice
= $5.29 , unit = $/Bu. ,
OUT = "C:\My NESUG 2011 - faking\faking 2011 State Guarantees and Benchmark Yields as of
&mydate..pdf" );

/* To output the PDF file with 22 maps in it. */
ods pdf close;
ods results on;
ods listing

/* End of: AO Crops all over June 17 2011 try.sas */

```

## Faking Esri ArcGIS maps in SAS®, continued

### APPENDIX 2. ACRE PRICE GUARANTEES THAT GET UPDATED IN A MASTER PROGRAM “CROPS ALL OVER.SAS”

2011 Revenue.xls [Compatibility Mode] - Microsoft Excel

Formulas Data Review View Add-Ins Acrobat

Wrap Text Merge & Center Alignment Number Conditional Formatting

been disabled Options...

EG	EH	EI	EJ	EK	EL	EM
			2011 ACRE Price Guarantee \$ per unit			
	Commodity	Unit				
	Barley	bushel	2.92			
	Canola	pound	0.1780			
	Corn	bushel	4.40			
	Crambe	pound	0.3600			
	Dry Peas	pound	0.0924			
	Flaxseed	bushel	10.17			
	Grain Sorghum	bushel	4.21			
	Large Chickpeas	pound	0.3035			
	Lentils	pound	0.2650			
	Long Grain Rice	pound	0.1200			
	Medium/Short Grain Rice	pound	0.1770			
	Mustard Seed	pound	0.2685			
	Oats	bushel	2.27			
	Peanuts	pound	0.220			
	Rapeseed	pound	0.2573			
	Safflower	pound	0.1870			
	Sesame Seed	pound	0.2930			
	Small Chickpeas	pound	0.2140			
	Soybeans	bushel	10.47			
	Sunflower Seed	pound	0.1945			
	Upland Cotton	pound	0.722			
	Wheat	bushel	5.29			

### APPENDIX 3. “AO SUPERMODULE GENERAL - REVENUE AND YIELD.SAS”

```

/*****
/* Name: AO Supermodule general - revenue and yield.sas */
/* Author: Anastasiya Osborne */
/* Date: June 17, 2011 */
/* GOAL: Fake ArcGIS maps for 22 crops. */
*****/

Data test1 (rename = (new&CROP.A = &CROP.A new&CROP.AY = &CROP.AY
                      new&CROP.I = &CROP.I new&CROP.IY = &CROP.IY
                      new&CROP.N = &CROP.N new&CROP.NY = &CROP.NY)) ;

  length state 5. abbr $2. ;
  Set xlsdta."Sheet1$"n (drop = state);

  state = input(code, 5.) ;
  abbr = fipstate(state);
  if state = . then delete ;

```

## Faking Esri ArcGIS maps in SAS®, continued

```

new&CROP.A = input(&CROP.A,$char15.); /* A - all revenue      */
old&CROP.A = &CROP.A ;

new&CROP.AY = input(&CROP.AY,$char15.); /* I - irrigated revenue */
old&CROP.AY = &CROP.AY ;

new&CROP.I = compress(input(&CROP.I,$char15.)); /*N - nonirrigated revenue */
old&CROP.I = &CROP.I ;

new&CROP.IY = input(&CROP.IY,$char15.); /* AY - all yield */
old&CROP.IY = &CROP.IY ;

new&CROP.N = input(&CROP.N,$char15.); /* IY - irrigated yield      */
old&CROP.N = &CROP.N ;

new&CROP.NY = input(&CROP.NY,$char15.); /* NY - nonirrigated yield      */
old&CROP.NY = &CROP.NY ;

drop &CROP.A &CROP.AY &CROP.I &CROP.IY &CROP.N &CROP.NY ;
run;

/* The SAS 9.3. version has the new LATON option allowing to use LAT and LONG variables directly
instead of X and Y. No conversion is needed. */

/* SAS 9.2 – data set with lat/long of center of each state*/

data centers (index=(state)); set maps.uscenter;
    where ocean eq 'N';
    x = long * constant('pi') / 180;
    y = lat * constant('pi') / 180;
    keep state x y; run;

/*****/
/* The plan:
- annotate data set
- read input data
- create the state abbreviation with the FIPSTATE function
- grab the lat/long of the state center from the CENTERS data set
- add the state abbreviation to the data set
- add non-missing crop values
/*****/

/* From SAS tech note, chapter 10: Create a label.
The value of the FUNCTION variable (LABEL) tells the program what to do. The values of the coordinate
variables X and Y combined with the values of the coordinate system variables HSYS, XSYS, and YSYS
tell where to do it. The values of the attribute variables STYLE, COLOR, TEXT, POSITION, and
SIZE tell how to do it. These variables specify the font (SWISSB), the color and text of the label, the
position of the label in relation to X and Y (centered on the point), and the size of the text.*/

data anno8;
    length text $45.    abbr $2. ;
    retain xsys ysys '2' hsys '3' function 'label' size 1.05 when 'a' cbox
'white';
    set test1 ;
    set centers key=state / unique;

```



## Faking Esri ArcGIS maps in SAS®, continued

```

color = 'black'; font = "Swissb" ;

abbr = fipstate(state);

if abbr = 'CT' then
do ;

    text = fipstate(state);      position = '1';    x=0.99*x ; y=0.98*y;
end;
output ;

if abbr = 'CT' then
do ;
    if &CROP.A ^= ' ' or &CROP.AY ^= ' ' then
    do;
        color = 'blue'; text = cat(&CROP.A) ; position = '4'; output;
        color = 'blue' ; text = cat(&CROP.AY) ; position =
            '7'; output;
    end ;
end ;

if abbr = 'MA' then
do ;
    text = fipstate(state); position = '1';    output;

    if &CROP.A ^= ' ' or &CROP.AY ^= ' ' then
    do;
        color = 'blue'; text = cat(&CROP.A) ; position = '4'; output;
        color = 'blue' ; text = cat(&CROP.AY) ; position = '7'; output;
    end ;
end ;

if abbr ^ in ( 'DE', 'RI', 'NH', 'CT', 'MA') then
do ;

    text = fipstate(state);      position = '2';    output;

    if &CROP.A ^= ' ' or &CROP.AY ^= ' ' then
    do;
        color = 'blue'; text = cat(&CROP.A) ; position = '5'; output;
        color = 'blue' ; text = cat(&CROP.AY) ; position = '8'; output;

        &CROP.I = compress(&CROP.I) ;
        &CROP.N = compress(&CROP.N) ;
    end ;

    if &CROP.I ^= ' ' and &CROP.N ^= ' ' then
    do ;
        separator = ' ' ;
        color = 'red'; text = catt(&CROP.I, ' / ', separator, &CROP.IY);
        position = '5'; output;
        color = 'black' ; text = catt(&CROP.N, ' / ', separator ,
            &CROP.NY ) ; position = '8'; output;
    end;

```

## Faking Esri ArcGIS maps in SAS®, continued

```

end ;

else if abbr in ('RI', 'NH') then
  do ;

    text = fipstate(state); position = '3';      output;
    color = 'blue'; text = cat(&CROP.A)      ; position = '6'; output;
    color = 'blue' ; text = cat(&CROP.AY)      ; position = '9'; output;
  end ;

else if abbr = 'DE' and crop ^= Corn then GOTO regularcrop;

  else if  abbr = "DE" and crop = Corn and &CROP.I ^= ' ' and &CROP.N ^=
  ' ' then
  do ;
    text = fipstate(state);      position = '3';      output;
    color = 'red'; text = catt(&CROP.I, ' / ', separator, &CROP.IY);
    position = '6'; output;
    color = 'black' ; text = catt(&CROP.N, ' / ', separator , &CROP.NY
    ) ; position = '9'; output;
  end;

regularcrop;

  else if abbr = 'DE' then
  do ;
    text = fipstate(state); position = '3';      output;
    color = 'blue'; text = cat(&CROP.A)      ; position = '6'; output;
    color = 'blue' ; text = cat(&CROP.AY)      ; position = '9'; output;
  end ;
run ;

data seeanno8 ;
  set anno8 (keep = abbr state text position &CROP.A &CROP.AY &CROP.I
&CROP.IY &CROP.N &CROP.NY) ;
run ;

/* Create a macro for Alaska's All Revenue and All Yield, using CALL SYMPUT */;

data forAK ;
  set anno8 ;

  if state = 2 ;

  AK_&CROP.A = cat(&CROP.A);
  AK_&CROP.AY = cat(&CROP.AY);
  call symput("AK_Allrev", AK_&CROP.A);
  call symput("AK_Allyld", AK_&CROP.AY);
run ;

/* make an annotation data set to add a small Alaska box in the upper left corner of the map. */
%annomac;

/* Now - use those macro variables for Alaska. */
data alaska;

```

## Faking Esri ArcGIS maps in SAS®, continued

```

length text $20;
retain xsys ysys '5' hsys '3' ;
%rect(-3,85,3,92,black,1,2);
function = 'label';
size = 1.25;
x = 0;
y = 91;
text = 'AK';
output;
color = 'blue';
x = 0;
y = 89;
text = "&AK_Allrev";
output ;
x = 0;
y = 87;
text = "&AK_Allyld" ;
output;
run;

data anno ;
  set anno8 (where = (state ^= 2)) ;
run ;

/*****
/* PROJECTING AND MAPPING
Once the annotation data set is built, the next steps are:
- combine it with the mapping coordinates for the boundaries
- project the combined data
- split the annotation data and boundary data into separate data sets
- create the map from the boundary, annotation, and response datasets.  */
*****/;

/* US map (no Alaska, Hawaii, Puerto Rico) */
data us1 ( rename = (state = badstate X = badX Y = badY ));
  set maps.states ;
  where state not in (2, 15, 72) and density <= 3;
run;

data us ;
  length state 5. X 6. Y 6. ;
  set us1 ;
  state = badstate ;
  X = badX ;
  Y = badY ;
run ;

/* combine map and annotate data step */
data withanno ;
  set us
      anno ;
run ;

/* project combined data */
proc gproject data=withanno

```

## Faking Esri ArcGIS maps in SAS®, continued

```

        out=pwithanno
        asis;
    id state;
run;

* separate map from annotation data set;
data usa (where=(xsys is missing))
      usa_anno (where=(xsys is not missing));

    set pwithanno ;
run;

*****
****      Preparation for maps              ****
*****
* reset/set graphics options;
goptions reset=all
        ftext= "Arial/bold"
        htext=3
        gunit=pct
        ROTATE=landscape /* The way it's printed on a page */
        ;
GSFMODE= append /* This is very important – this controls whether maps replace one
another, or append to one PDF file. */

legend1 /* fake legend in the lower left corner of the map.*/
label=(position=top
h= 2 /*1.75 */
j=1 'State Abbreviation'
c=blue
j=1 'BLUE: All Revenue'
j=1 'BLUE: All Yield'
j=1 ' '
c=black
j=1 'State Abbreviation'

j=1 c=red 'RED: Irrigated Revenue / Irrigated Yield'
j=1 c=black 'BLACK: NonIrrigated Revenue / NonIrrigated Yield'
j=1 ' '
)
value=none
shape=bar(1e-10,1e-10)pct
origin=(5,5)pct
mode=share
;

pattern v=e c=black; /* map will have empty states with black outline */

/* white space around map */
title1 ls=1;
title2 a=90 ls=2;
title3 a=-90 ls=2;

ods listing close;
ods results off;

```

## Faking Esri ArcGIS maps in SAS®, continued

`ods pdf file= OUT notoc ;` /\* The NOTOC\_DATA option suppresses any hidden RTF control strings which are put into the table headers just in case one needs to build a table of contents in Word. This option was introduced in SAS 9.1.3, Service Pack 4. \*/

`ods escapechar='^';` /\* This means that anytime ODS encounters the caret (^) it will interpret whatever follows the caret (^) to be a sequence or string of control characters that will have an impact on the final output file, as created by ODS.\*/

```
proc gmap data=usa
      map=usa
      annotate=usa_anno;
  id state;
  choro state / levels= 1 legend=legend1 annotate=alaska;

  note height=14 pt f = swissb
    j=c "&CR Preliminary ACRE Program Guarantees in ($/Acre) and" ls=1
    j=c "Benchmark State Yields in (&state_yield) for &Iyear Crop Year"
    j=c h=2 "&guarprice Guarantee Price in (&unit)";

  note height=11 pt f = swissb move=(55,1) color=black
    "Latest Update: &mydate";
run;
quit;
```

/\* End of "AO Supermodule general - revenue and yield.sas \*/

## APPENDIX 4. EARLIER CODE WITH ARRAYS THAT WORKED FOR MOST OF THE CROPS, BUT NOT FOR ALL – AND NEEDED TO BE REPLACED.

```
Data cleanit (rename = (new&CROP.A = &CROP.A new&CROP.AY = &CROP.AY
                        new&CROP.I = &CROP.I new&CROP.IY = &CROP.IY
                        new&CROP.N = &CROP.N new&CROP.NY = &CROP.NY)) ;
  length state 5. abbr $2. ;
  set xlsdta."Sheet1$"n (drop = state);

  state = input(code, 5.) ;
  abbr = fipstate(state);
  if state = . then delete ;

  array agvars (6) &CROP.A &CROP.AY &CROP.I &CROP.IY &CROP.N &CROP.NY ;
  array oldagvars (6) $ old&CROP.A old&CROP.AY old&CROP.I old&CROP.IY
    old&CROP.N old&CROP.NY ;
  array newagvars (6) $ new&CROP.A new&CROP.AY new&CROP.I new&CROP.IY
    new&CROP.N new&CROP.NY ;

  do i = 1 to 6 ;
    newagvars(i) = input(agvars(i), $char15.);
    oldagvars (i) = newagvars(i);
  end ;
  /* vars:
    A - all revenue, I - irrigated revenue, N - nonirrigated revenue
    AY - all yield, IY - irrigated yield, NY - nonirrigated yield */

  drop &CROP.A &CROP.AY &CROP.I &CROP.IY &CROP.N &CROP.NY ; Run;
```