

Paper 269-2012

## Sparklines in SAS® DSGI

Murphy Choy, School of Information System, SMU, Singapore

### ABSTRACT


Sparkline is one of the latest visualization techniques that analysts are using to assist them in visualizing information. A pet invention by the famous visualization expert Edward Tufte, this form of visualization has incorporated many of this key ideas in the use of chart and diagrams to reinforce certain concepts and ideas. Sparkline can be implemented in SAS via certain ways. In this paper, we will illustrate the ease of creating sparklines with SAS DSGI techniques.

### INTRODUCTION

Sparkline as a visualization tool is first proposed by Edward Tufte in his book the Visual Display of Quantitative Information. Edward Tufte emphasized in his book the importance of proper data visualization and what are the key drivers that impact the visual display of information. Many concepts that he mentioned were incorporated into the design of the Sparkline marking it one of the most intense form of visualization. There have been widespread applications of the tool especially in the stock markets and time series intensive industries. In this paper, we will discuss how Sparklines can be drawn in SAS DSGI and how they are used as intensive visualizations.


### DERIVATION OF SPARKLINES

Sparklines have a very simple structure. They are essentially lines that depict the trend of the time series. As opposed to conventional time series, the Sparklines have no axes and relative measures. This reduces the overall information on the data but does further to emphasize the key elements from the data. This creates something that is succinct and easy to remember. Below is an example of a Sparkline isolated.



Combining a few Sparklines together form an intensive chart of information.

Sparklines in SAS® DSGI, continued



To create a sparkline, one only needs to find out the maximum and minimum of a trend and converting them to a standardized scale which is use in the plotting of the lines. Once the lines have been plotted some additional details may be added about the stock or the closing price. The macro needed to create the Sparklines can be found in Appendix A. For multiple charts, one can also resort to the use of the PROC GREPLAY to create the chart which can be found in Appendix B.

## CONCLUSION

Sparklines are simple visualization tools for intense time series visualization.

## CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the authors at:

Name: Murphy Choy  
 Enterprise: School of Information Systems, Singapore Management University  
 Address: 80 Stamford Road  
 City, State ZIP: Singapore 178902  
 Work Phone: +65-92384058  
 E-mail: goladin@gmail.com/murphychoy@smu.edu.sg

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. © indicates USA registration.

Other brand and product names are trademarks of their respective companies.

Sparklines in SAS® DSGI, continued

## APPENDIX A

```
%MACRO SPARKSLINES (INPUT, X, Y, XCOR, YCOR) ;

/*SORTING THE DATA BY THE DATE VARIABLE*/

PROC SORT DATA = &INPUT; BY &X; RUN;

/*INITIAL DATA PREPARATION STEP*/

DATA TEMP;

    /*FIRST DOW LOOP*/

    DO UNTIL (EOF) ;

        SET &INPUT END = EOF;

        /*SETTING THE MAXIMUM AND MINIMUM FOR Y VARIABLE*/
        MAXVAL = MAX (&Y, MAXVAL) ;
        MINVAL = MIN (&Y, MINVAL) ;

        /*SETTING THE MAXIMUM AND MINIMUM FOR X VARIABLE*/
        MAXVAL2 = MAX (&X, MAXVAL2) ;
        MINVAL2 = MIN (&X, MINVAL2) ;

    END;

    /*INTERMEDIATE WORKINGS*/

    /*EXTRACTING THE RANGE OF VALUES*/
    RANGE = MAXVAL - MINVAL;
    RANGE2 = MAXVAL2 - MINVAL2;

    /*SETTING MULTIPLIER VALUE*/
    TEST = RANGE; TEST2 = RANGE2;
    I = 1; J = 1;

    /*CREATING A Y LIMIT CAP OF 10*/
    DO WHILE (TEST > 70) ;
```

Sparklines in SAS® DSGI, continued

```
        I = I+1;
        TEST = RANGE/I;

    END;

    /*CREATING AN X LIMIT CAP OF 50*/
    DO WHILE (TEST2 > 150);

        J = J+1;
        TEST2 = RANGE2/J;

    END;

    /*SETTING THE RANGE LIMIT ON GRAPH*/
    CALL SYMPUT ('L1',TEST);
    CALL SYMPUT ('L2',TEST2);

    /*SECOND DOW LOOP*/
    DO UNTIL (EOF2);

        SET &INPUT END = EOF2;

        /*FLAGGING VALUE*/
        ORIG = &Y;

        /*CREATING THE NEW VALUES*/
        &Y = (&Y-MINVAL)/I;
        &X = (&X-MINVAL2)/J;

        /*CREATING LAGGED VALUES*/
        C = LAG (&X); D=LAG (&Y);
        OUTPUT;

    END;

RUN;

/*****/

PROC SQL NOPRINT;
```

## Sparklines in SAS® DSGI, continued

```

/*ESTIMATING THE MEAN AND STANDARD DEVIATIONS*/
SELECT MEAN(&Y),STD(&Y) INTO :CA1,:CA2 FROM TEMP;

QUIT;

/*****/

/*DSGI DRAWING SECTION*/

DATA _NULL_;

/*setting the data*/
SET TEMP END = EOF;

IF _N_ = 1 THEN DO;

    /*initialization of the graph plotting area*/
    RC = GINIT();
    RC = GRAPH('CLEAR',"&INPUT");
    /*setting the overall domain*/
    RC = GSET('WINDOW',1,0,0,200,100);

    /*creating the container boxes*/
    RC=GSET('LINWIDTH',1);
    RC = GSET('LINCOLOR',3);

    RC = GDRAW('BAR',&XCOR-5,&YCOR-5,&XCOR+35+&L2,&YCOR+10+&L1);

END;
ELSE IF NOT EOF THEN DO;

    /*line qualities declarations*/
    RC = GSET('LINCOLOR',1);
    RC = GSET('LINTYPE',1);
    RC=GSET('LINWIDTH',2);

    /*drawing the lines*/
    RC = GDRAW('LINE',2,&XCOR+&X,&XCOR+C,&YCOR+&Y,&YCOR+D);

END;

```

Sparklines in SAS® DSGI, continued

```
IF EOF THEN DO;

    /*creating the final dot*/
    RC=GSET('LINWIDTH',3);
    RC = GSET('LINCOLOR',5);
    RC = GDRAW('LINE',2,&XCOR+&X,&XCOR+C,&YCOR+&Y,&YCOR+D);

    /*outputting the words and price value*/
    RC = GSET('TEXHEIGHT', 5);
    RC = GSET('TEXTFONT', 'SWISSL');
    K = COMPBL("PRICE : "||ORIG);
    RC = GDRAW('TEXT', &XCOR+5+&L2, &YCOR+&L1, "STOCK: &INPUT");
    RC = GDRAW('TEXT', &XCOR+5+&L2, &YCOR-5+&L1, K);

    /*create the entire chart*/

    RC = GRAPH('UPDATE');
    RC = GTERM();

END;

RUN;

%MEND;
```

Sparklines in SAS® DSGI, continued

## APPENDIX B

```
/*MULTIPLE INSETS*/

DATA _NULL_;

        RC = GINIT();
        RC = GRAPH('CLEAR');

/*POSITIONS*/
DSGI=GSET('VIEWPORT', 1, 0.00, 0.60, 0.60, 1.00);
DSGI=GSET('VIEWPORT', 2, 0.40, 0.60, 1.00, 1.00);
DSGI=GSET('VIEWPORT', 3, 0.00, 0.10, 0.60, 0.50);
DSGI=GSET('VIEWPORT', 4, 0.50, 0.00, 1.00, 0.60);

/*SETTING THE GRAPHS*/
RC=GSET('TRANSNO', 1);
RC=GRAPH('INSERT', 'ASIA');

RC=GSET('TRANSNO', 2);
RC=GRAPH('INSERT', 'BIO');

RC=GSET('TRANSNO', 3);
RC=GRAPH('INSERT', 'China');

RC=GSET('TRANSNO', 4);
RC=GRAPH('INSERT', 'Gallant');

RC = GRAPH('UPDATE');
RC = GTERM();

RUN;

/*PLAY THE ITEMS*/
PROC GREPLAY IGOUT= GSEG NOFS TC=SASHELP.TEMPLT;
TEMPLATE L2R2;
TREPLAY 1:ASIA 2:BIO 3:China 4:Gallant;
RUN;
QUIT;
```